
==================PhaseFrequencyDetector_Tests==================

A CIRCUIT THAT DEFINITELY NEEDS BREAK BEFORE MAKE LOGIC TO CONVERGE.

 

 

 

PhaseFrequencyDetector_Tests
*
*        ^ Vcc
*       /_\   __________                      ___
*        |___| D      Q |____________________|DWN|
*            |    1     |                 |  |___|
*  ___       |          |                 |
* |REF|______|CLK RST   |                 |
* |___|      |__________|     _ RS  ___   |
*                  |     RST / |   /   |__|
*        ^ Vcc     |________/4 |__/ 3  |        2.23.11_10.13AM  dsauersanjose@aol.com 
*       /_\   _____|____    \  |  \    |__      Don Sauer        http://www.idea2ic.com/
*        |___| D  RST   |    \_|   \___|  |
*            |          |   tau           |
*  ___       |    2     |                 |   ___
* |VCO|______|CLK     Q |_________________|__|UP |
* |___|      |__________|                    |___|
*

*=========Create_Signal==================
VTime     VTime  0      DC     0     PWL(   0      0    1      1)
VCC       VCC    0      DC     1
Vfreq1    Vfreq1 0      DC     1000
Vfreq2    Vfreq2 0      DC     1100



Vphas2    Vphase 0      DC     0
BREF      REF    0      V      =     u(cos(6.2831*V(VFreq1)*V(VTime)))
BVCO      VCO    0      V      =     u(cos(6.2831*V(VFreq2)*V(VTime)+V(Vphase)))
*XDRS_FF  D      CLK    RST    Q     DRS_FF
XDRS_FF1  VCC    REF    RST    DWN   DRS_FF
XDRS_FF2  VCC    VCO    RST    UP    DRS_FF
BAND3     RS     0      V      =     u( u(v(DWN )-.5)*u( v(UP) -.5   ) -.5)
XD_tau4   RS     RST                 D_tau
.control
set       pensize = 2
tran      1u    20m     0      1u
plot      ref+.1 vco-1.1 up-2.2 dwn-3.3 title REFfreq_GT_VCOfreq
alter     Vfreq1 dc =    1100
alter     Vfreq2 dc =    1000
tran      1u     20m     0      1u
plot      ref+.1 vco-1.1 up-2.2 dwn-3.3 title REFfreq_LT_VCOfreq
alter     Vfreq1 dc =    1000
alter     Vfreq2 dc =    1000
tran      1u     20m     0      1u
plot      ref+.1 vco-1.1 up-2.2 dwn-3.3 title REFfreq_EQ_VCOfreq
alter     Vphas2 dc =   -1.57
tran      1u     20m     0      1u
plot      ref+.1 vco-1.1 up-2.2 dwn-3.3 title VCOPhase_EQ_neg90
alter     Vphas2 dc =    1.57
tran      1u     20m     0      1u
plot      ref+.1 vco-1.1 up-2.2 dwn-3.3 title VCOPhase_EQ_90
alter     Vphas2 dc =    3.14
tran      1u     20m     0      1u
plot      ref+.1 vco-1.1 up-2.2 dwn-3.3 title VCOPhase_EQ_180
.endc

*=========D_tau===========================================
*
*  ___   |\ D1         VLP  |\   ___
* |IN |__|1\_/\  /\  /\_____|2\_|OUT|
* |___|  | /   \/  \/  _|_  | / |___|
*        |/    RLP     ___  |/
*                 CLP  _|_
*                      ///
*

.SUBCKT   D_tau  IN     OUT 
 B1       D1     0      V =    u( v(IN  )-0.5 )
 RLP      D1     VLP    1K
 CLP      VLP    0      1n
 B2       OUT    0      V =    u( v(VLP  )-0.5 )
.ENDS     D_tau   

*=========D_reset_FF===========================================
*
*  ___  |\ D1/     D2   |\    ___
* |D  |_|1\_/  _________|6\__|Q  |
* |___| | / S2  _|_  |  | /  |___|
*       |/      ___  |  |/                  __________
*             C1 |   |    /   ___        __| D      Q |__
*               _|_  |___/  _|  _|_        |          |
*               ///      S1     ///        |          |
*                                        __|CLK RST   |__
*  ___   |\            VBF2 ____           |__________|
* |CLK|__|2\________________|   \ CS1            |
* |___|  | / |          VLP2| 4  \___
*        |/  |/\  /\  /\____|    /
*               \/  \/  _|_ |___/
*               RLP2    ___
*                  CLP2 _|_
*                       ///
*  ___   |\    VBF3         ____
* |RST|__|3\________________|   \ CS2
* |___|  | / |          VLP3| 5  \___
*        |/  |/\  /\  /\____|    /
*               \/  \/  _|_ |___/
*               RLP3    ___
*                  CLP3 _|_
*                       ///
.SUBCKT   DRS_FF  D     CLK    RST     Q 
 B1       D1      0     V =    u( v(D   )-0.5 )
 S1       D1      D2    CS1    0        SW 
 S2       D2      0     CS2    0        SW 
 C2       D2      0     10n
 R1       D2      0     100G   
 B6       Q       0     V =    u( v(D2   )-0.5 )
 B2       VBF2    0     V =    u( v(CLK ) -0.5 )
 RLP2     VBF2    VLP2  1K
 CLP2     VLP2    0     .1n
 BAND4    CS1     0     V =    u( u(v(VBF2 )-0.5)*u(.5 -v(VLP2 ) ) -.5)
 B3       VBF3    0     V =    u( v(RST    )-0.5)
 RLP3     VBF3    VLP3  1K
 CLP3     VLP3    0     .1n
 BAND5    CS2     0     V =    u( u(v(VBF3 )-0.5)*u(.5 -v(VLP3 ) ) -.5)
.ENDS     DRS_FF 

.MODEL    SW      SW(    VT=.5 VH=.1  RON=1 ROFF=100MEG)

.end


